

Justin Yatsunashi and Daniel Noronha

Professor Fuja

Embedded Systems

20 April 2022

## **EE 10115 Temperature Regulator Project Report**

### **Executive Summary**

The project being demonstrated is a temperature regulator which checks the current ambient temperature and adjusts the speed of a DC fan accordingly. It has two modes: Manual and Automatic. In the Manual mode, the user can control the fan speed directly. In the Automatic mode, the system will calculate the error between the current and user-set temperature, then will use this value to determine the fan's speed. To switch between these two modes, the user pushes a button. The system will also include an LCD to display the set temperature or fan speed and the temperature of the surroundings.

### **Background**

The device is extremely useful for regulating the temperature of a room. In the automatic mode, it will spin faster in order to cool the room to the set temperature, then it will slow down to maintain that temperature. Meanwhile, in the manual mode, the user can simply set how fast they want the fan to spin in order to have the desired cooling effect. This device could be useful as an air conditioner in different buildings and rooms in order to regulate and maintain a comfortable temperature. For instance, it would be a very useful system for dorms without air conditioning at Notre Dame. Since it runs on battery power, it is very portable and can be used as a source of cooling in rooms without mains electricity or outdoors in hot weather.

The idea was an original design that was created for this project, but its functionality is similar to that of a standard air conditioning system (temperature sensing and actuation). Like an air conditioner, the temperature regulator allows the user to set the temperature and adjusts the cooling effect based on that setting. This means that the fan will cool the room to the desired temperature just like an air conditioning unit would do. Devices like this exist in the marketplace, but most of them require a lot of power to run and hence can only be used with mains (AC) electricity. An example of a similar embedded system would be a computer cooling

fan used as a CPU cooler or case fan. This project device is meant for personal use by individuals as its low power consumption lets it run on battery power and makes it portable.

### **Functional Description and Product Operation**

The temperature regulator has a manual and automatic mode which the user can toggle between by pushing a button. In both modes the LCD will show the current ambient temperature in degrees Celsius and Fahrenheit. The currently active mode is denoted by the letter “M” for manual and “A” for automatic which can be found at all times in the bottom right corner of the LCD display. The LCD contrast can be adjusted using a second (dedicated) potentiometer knob.

In the manual mode, the user turns a potentiometer knob to set the desired fan speed as a percentage (10%-100%), which is shown on the second line of the LCD.

In the automatic mode, the user turns the same potentiometer knob to set the desired temperature. The currently set temperature is shown on the second line of the LCD in degrees Celsius and Fahrenheit. The user can set a temperature between 21°C (70°F) and 27°C (81°F). This range was chosen because the fan can cool the surroundings to achieve the user’s set temperature within a reasonable amount of time (for demonstration purposes).

### **Hardware**

The temperature regulator employs a variety of different sensors and actuators.


Sensors:

- Thermistor (temperature sensor)
- Potentiometer
- Push Button

Actuators:

- 120mm 4-pin PWM RGB Fan (12V DC motor)
- 16x2 LCD

## Hardware Setup/Circuit Diagram



The thermistor connected to pin A3 is used to determine the current ambient temperature so the fan speed can be adjusted accordingly. The potentiometer on pin A0 is used to either set the desired temperature (in the automatic mode) or to set the fan speed (in the manual mode) by turning the knob. The push button conveniently switches the mode between automatic and manual by triggering an interrupt service routine on pin 2. The 120mm 4-pin PWM RGB Fan (12V DC motor) cools the surroundings and was chosen because its speed can be controlled directly with a PWM signal from digital pin 9, while the LCD displays information in a user-friendly manner. The fan's power pins are connected to an external circuit with 18V (from two 9V batteries in series) and to the Arduino circuit through its PWM pin.

## Software

The code reads the three sensors, and uses the analog input (or digital in the case of the push button) to determine what signal to send to the two actuators.

Before the setup function, the LiquidCrystal.h header file is imported and some constants relating to fan speed and pin selection are set.

### **setup ()**

The setup function sets the fan PWM pin mode to output and attaches a rising interrupt function (modeSwitch) to the push button on pin 2. Additionally, it sets the size of the LCD to 16x2 and displays a start screen with the project name and the names of its creators for 2.5 seconds before executing the rest of the code. Finally, it sets the initial mode as “manual.”

### **loop()**

The loop function first clears the LCD then uses the analogRead() in order to obtain sensor values from the potentiometer and the thermistor. The program then uses the map function to convert the potentiometer value to a fan speed between 26 (10%) and 255 (100%). The fan speed is calculated as a percentage of 255. The user-defined function therm2cel converts the thermistor value to a temperature in degrees Celsius. The current temperature is printed to the LCD in degrees Celsius and Fahrenheit and the LCD cursor is set to the next line.

Since the mode is initially set to manual, the program will execute the else condition block, which instructs the fan to spin at the user-set speed using the analogWrite() function. The fan speed percentage is printed to the LCD display along with the letter “M” to denote that the system is in manual mode.

If the push button is pressed, the mode program will execute the if (automatic) condition block. It will first display the set temperature in degrees Celsius and Fahrenheit while also printing “A” for automatic. If the ambient temperature is more than 1°C above the set temperature, the fan speed is set to 100% (255). If it is within 1°C above the set temperature, the fan is set to 50% (127). If the current temperature is less than the set temperature, the fan will remain at 10% (26) to maintain the current temperature. Finally, the program will delay for a half-second before looping.

### **therm2cel()**

This function takes the value read from the thermistor (pin A3), and converts it to millivolts (0V to 5000V) using the map function. Then, it determines the Celsius temperature using an equation based on the temperature-resistance relationship of the TMP36 thermistor:  $\text{temp} = (\text{mV} - 500.0) / 10$ . The function returns the temperature in degrees Celsius.

## **modeSwitch()**

This function has no inputs or outputs. It is an interrupt service routine that toggles the value of the volatile boolean 'automatic' between false and true as long as it has been at least 200 milliseconds between two consecutive button pushes (to eliminate button bounce).

## **Program Flow**

After initializing global variables, the program performs the setup() function and then proceeds to the loop() function. In the loop function(), the user-defined function therm2cel() is called to determine the current ambient temperature, before the program loops again. Pushing the button triggers the modeSwitch() interrupt service routine which toggles the system mode as needed.

## **Project Post-Mortem**

The final product was almost the same as described in the initial proposal. The only change made was to display temperatures in Celsius and Fahrenheit instead of only in Celsius. Overall, the most interesting part of the project is its real-use application. The temperature regulator's function is basically the same as that of an air conditioning unit (except that it only uses a fan), and it was interesting to see how such a unit actually works. The system effectively uses a feedback mechanism to maintain the temperature that the user sets in the automatic mode and can lower the temperature quite quickly for a single fan. The most difficult part of the project was actually fitting all of the components and wires onto a single breadboard, and eventually the button had to be put onto a separate board because there was not enough room on the first one. If the project were to be done again, the system would be modified so that in automatic mode, the fan turns off completely when the temperature drops below the set temperature instead of spinning at idle (10% speed). The fan itself currently does not spin at less than 10% even with a PWM duty cycle of 0 (a design limitation). The provided MOSFET could not be used to stop the fan either because with a gate voltage of 5V, the RDS(on) is too high resulting in a very high voltage drop across the transistor (~15V/18V) when in the on state (which leaves too little voltage for the fan to run). This issue could be fixed by using a transistor with a lower RDS(on) value. Also, in automatic mode, a proportional control algorithm could be used instead of stepping between 0%, 50%, and 100% fan speed.

## Appendix

```
1  /*EE_TempRegulator_Project.ino
2  This program contains the code for a temperature regulator
3  which checks the current ambient temperature and adjusts
4  the speed of a DC fan accordingly.
5  It has two modes: Manual (user sets fan speed)
6  and Automatic (user sets desired temperature).
7  Sensors:
8  - Thermistor (temperature sensor)
9  - Potentiometer
10 - Push Button
11 Actuators:
12 - 120mm 4-pin PWM RGB Fan (12V DC motor)
13 - 16x2 LCD
14
15 Date: 04/20/2022
16 Author: Daniel Noronha & Justin Yatsushashi
17 EE 10115 Temperature Regulator Project Code
18 */
19
20 // Include LCD header
21 #include <LiquidCrystal.h>
22 // Set mode to manual by default but keep it volatile
23 // Can be accessed by interrupt
24 volatile bool automatic = false;
25 // Hardware limitations mean PWM duty cycle < 26 has no effect.
26 const int min_PWM_speed = 26; // ~10% fan speed (26/255)
27 // Define hardware pins
28 const int fan_PWM_pin = 9;
29 const int button_pin = 2;
30
31 LiquidCrystal lcd(11, 12, 4, 5, 6, 7);
32
33 void setup() {
34   // Setup code which runs on startup only
35   // Set fan digital PWM pin mode to OUTPUT
36   pinMode(fan_PWM_pin, OUTPUT);
37   // Attach rising interrupt to push button (on pin 2 = 0)
38   attachInterrupt(button_pin-2, modeSwitch, RISING);
39   // Initialize 16x2 LCD
40   lcd.begin(16, 2);
41   // Display Startup Message for 2.5 seconds
42   lcd.setCursor(1,0);
43   lcd.print("Temp Regulator");
44   lcd.setCursor(0,1);
45   lcd.print("By Daniel&Justin");
46   delay(2500);
47 }
48
49 void loop() {
50   // Main program loop (indefinite)
51   // First clear the LCD
52   lcd.clear();
53   // Read thermistor temperature value (unmapped)
```

```

54 int sensorVal = analogRead(A3);
55 // Read potentiometer value
56 int potVal = analogRead(A0);
57
58 // Use therm2cel function defined below to map thermistor value to
temperature
59 float temp = therm2cel(sensorVal); // temperature in °C
60 // Display current temperature on first line
61 lcd.print(temp);
62 lcd.print("°C");
63 lcd.print(" ");
64 lcd.print(temp*1.8+32); // °C to °F
65 lcd.print("°F");
66 lcd.setCursor(0,1);
67 // If the mode is automatic:
68 if (automatic) {
69     // Map knob setting to temperature range (21°C-27°C)
70     int settemp = map(potVal, 0, 1023, 21, 27);
71     // Display set temperature on second line
72     lcd.print("Set: ");
73     lcd.print(settemp);
74     lcd.print("°C");
75     lcd.print(int(settemp*1.8+32)); // °C to °F
76     lcd.print("°F A"); // Show selected mode as Automatic (A)
77     // If the current temperature is less than 1°C above the
78     // set temperature, then set fan speed to 50% (127/255)
79     if (int(temp) == settemp) analogWrite(fan_PWM_pin, 127);
80     // Otherwise, if the current temperature is more than 1°C above the
81     // set temperature, then set fan speed to 100% (255/255)
82     else if (temp > settemp) analogWrite(fan_PWM_pin, 255);
83     // If the current temperature is less than the set temperature,
84     // then set the fan to idle (10%)
85     else analogWrite(fan_PWM_pin, 0);
86 }
87
88 else {
89     // Manual Mode
90     // Map knob setting to PWM fan speed (26-255)
91     int fanSpeed = map(potVal, 0, 1023, min_PWM_speed, 255);
92     // Express PWM fan speed as a percentage (10%-100%)
93     int percent = (fanSpeed/255.0) * 100;
94     // Send PWM signal to fan
95     analogWrite(fan_PWM_pin, fanSpeed);
96     // Display user-set fan speed percentage on second line
97     lcd.print(percent);
98     lcd.print("%");
99     lcd.setCursor(15,1);
100    lcd.print("M"); // Show selected mode as Manual (M)
101 }
102 delay(500);
103 // Wait half a second to let user read display before clearing
104
105 }

```



```

106
107 float therm2cel(int thermval) {
108     // User-defined function to map thermistor value to temperature (°C)
109     // Map thermistor value to millivolts (0-5mV)
110     float mVolts = map(thermval, 0, 1023, 0, 5000);
111     // Use linear equation to convert voltage to Celsius temperature
112     float temp = (mVolts - 500.0) / 10.0;
113     return temp;
114 }
115
116 void modeSwitch() {
117     // Interrupt Service Routine to toggle system mode selection (A/M)
118     // Checks how long it has been since last processed interrupt
119     // If it has been less than 0.2s, do not process interrupt
120     // Eliminates button bounce
121     static unsigned long last_processed_interrupt_time = 0;
122     unsigned long current_interrupt_time = millis();
123     if ((current_interrupt_time - last_processed_interrupt_time) > 200) {
124         // Bit-flip boolean value of global volatile variable automatic (0/1)
125         automatic = !automatic; // false=>true or true=>false
126     }
127     last_processed_interrupt_time = current_interrupt_time;
128 }
129
130 // End of Program //

```